Picturing the Reeb Foliation

Jonathan Love

Introduction

The Hopf fibration is too mainstream. Instead of covering the 3-sphere with smooth curves, let's cover it with surfaces.

The Reeb foliation of the 3-sphere has a relatively simple explanation:
- Embedding a central torus C in the 3-sphere partitions it into two solid tori
- Within each solid torus, the leaves are all identical, just rotated copies of each other. Each can be interpreted as the surface of revolution of a curve that starts with a vertical tangent then tends to a horizontal asymptote, but wrapped around itself inside the torus.

First Parametrization: Naïve Approach

Second Parametrization

Let's look for a more natural parametrization of the Reeb foliation by considering where it lies in $S^3 \subset \mathbb{R}^4$. Setting up:

```
F1[x_] := (2/π ArcTan[x])^(3/4);
(* The square root looks too blunt-ended in this case *)
qual = 6;
p = 5; (* torus ratio *)
```

The torus naturally embeds in S^3 as $\frac{1}{\sqrt{2}}(\cos\theta, \sin\theta, \cos\psi, \sin\psi)$, i.e. $x^2 + y^2 = z^2 + w^2$, and divides the sphere into solid tori $x^2 + y^2 > z^2 + w^2$ and $x^2 + y^2 < z^2 + w^2$. Thus the 3-sphere will be parametrized by level sets of $\frac{x^2+y^2}{z^2+w^2}$.

Stereographic projection from (0,0,0,1) maps a point (x', y', z', w') on S^3 to $\left(\frac{x'}{1-w'}, \frac{y'}{1-w'}, \frac{z'}{1-w'}\right)$ in R^3. Given $r = \frac{x^2+y^2}{z^2+w^2}$, we know $(1+r)(z^2+w^2) = 1$, hence $z^2 + w^2 = \frac{1}{1+r}$, so $x^2 + y^2 = \frac{r}{1+r}$. So the points on a "torus of ratio r" satisfy $\frac{1}{\sqrt{1+r}}\left(\sqrt{r}\cos\theta, \sqrt{r}\sin\theta, \cos\psi, \sin\psi\right)$. This is pushed forward to the following, which we use to plot roughly the same picture as above.

```
TorusPoints2[θ_, ψ_, r_] := {Sqrt[r] Cos[θ]/(Sqrt[1 + r] - Sin[ψ]), Sqrt[r] Sin[θ]/(Sqrt[1 + r] - Sin[ψ]), Cos[ψ]/(Sqrt[1 + r] - Sin[ψ])};
```

And now we make the picture. One thing to point out is the fact that I'm plugging in cubes or fifth powers. The reason for this is that originally, the rendered picture looked terrible, because this parametrization moves incredibly quickly away from $r = 0$. No matter how high I increased MaxRecursion or PlotPoints, *Mathematica* simply didn't use enough points to make the generated surfaces smooth (or by the time it sampled enough points, the execution time was far too long). By composing with a function that is very flat around 0, I force *Mathematica* to sample many points where needed, and fewer elsewhere.

```
leaves2 = {0, π/2, 1.1 π, 1.25 π, 1.38 π, 1.45 π, 1.495 π, 1.55 π, 1.62 π, 1.75 π};
outers2 =
  ParametricPlot3D[TorusPoints2[θ, β^5 - #, p F1[β^5]], {θ, 0, π}, {β, 0, (2 π)^(1/5)},
     Mesh → None, PlotStyle → {Hue[Mod[# - 1.49 π, 2 π] / 30 + 1 / 4],
       Specularity[White, 100]}, MaxRecursion → qual] & /@ leaves2;
insides2 = Table[ParametricPlot3D[TorusPoints2[θ^3 - k, α, p / F1[θ^3]],
    {θ, 0, π^(1/3)}, {α, π/2 + .2, 3 π/2 + .2}, Mesh → None,
    PlotStyle → {Hue[k / 30], Specularity[White, 100]},
    MaxRecursion → qual], {k, 0, 2 π - π / 6, π / 6}];
sampleinside2 = ParametricPlot3D[TorusPoints2[θ^3 - π / 6, α, p / F1[θ^3]],
   {θ, 0, (3 π / 2)^(1/3)}, {α, -π, π}, Mesh → None,
   PlotStyle → {Hue[Pi / 90], Specularity[White, 100]}, MaxRecursion → qual];
Ctorus2 = ParametricPlot3D[TorusPoints2[θ, α, p], {θ, 0, 2 π}, {α, π/2 + .2, 3 π/2 + .2},
   Mesh → None, PlotStyle → {Blue, Specularity[White, 100]}, MaxRecursion → qual];
```

```
Show[{insides2, outers2, sampleinside2, Ctorus2},
  PlotRange → {{-10, 4}, {-2, 10}, {-8, 4}}, Boxed → False,
  Axes → False, Background → Gray, ImageSize → {1296, 768},
  ViewVector → {{12, -24, 12}, {-0.9, 0, 0}},
  ViewAngle → π/24, ViewVertical → {0, 0, 1}] // Rasterize
```



MUCH simpler than last time!

Third Parametrization (and solving a challenge question)

Maybe by using a different projection of the torus we can obtain another, more symmetrical picture? Let's rotate the main torus so that the point $\left(0, \frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right)$ is sent to (0, 0, 0, 1); that is, (x, y, z, w) is sent to $\left(x, \frac{1}{\sqrt{2}}(y-w), z, \frac{1}{\sqrt{2}}(y+w)\right)$. Thus, the point on the "torus of ratio r" $\frac{1}{\sqrt{1+r}}\left(\sqrt{r}\cos\theta, \sqrt{r}\sin\theta, \cos\psi, \sin\psi\right)$ is sent to $\frac{1}{\sqrt{1+r}}\left(\sqrt{r}\cos\theta, \frac{1}{\sqrt{2}}\left(\sqrt{r}\sin\theta-\sin\psi\right), \cos\psi, \frac{1}{\sqrt{2}}\left(\sqrt{r}\sin\theta+\sin\psi\right)\right)$, which is then projected to $\mathbb{R}^3$:

```
TorusPoints3[θ_, ψ_, r_] := {Sqrt[2 r] Cos[θ]/(Sqrt[2 + 2 r] - Sqrt[r] Sin[θ] - Sin[ψ]),
   (Sqrt[r] Sin[θ] - Sin[ψ])/(Sqrt[2 + 2 r] - Sqrt[r] Sin[θ] - Sin[ψ]), Sqrt[2] Cos[ψ]/(Sqrt[2 + 2 r] - Sqrt[r] Sin[θ] - Sin[ψ])};
```

Incidentally, this answers another of the challenge problems: Find a parametrization of the torus with an obvious nontrivial 4-fold symmetry!

```
ParametricPlot3D[TorusPoints3[θ, ψ, 1], {θ, 0, 2 π}, {ψ, 0, 2 π},
 RegionFunction → Function[{x, y, z}, Max[Abs /@ {x, y, z}] < 3],
 Mesh → {{π/2, π, 3 π/2}}, BoundaryStyle → Thin, Boxed → False, Axes → False]
```



And now, the pretty picture.

```
tops = Table[ParametricPlot3D[TorusPoints3[θ^3 - k, α, 1 / F[θ^3]],
    {θ, 0, (4 π / 3)^(1/3)}, {α, π / 2, 3 π / 2}, Mesh → None,
    PlotStyle → {Hue[Mod[k - 3 π / 2, 2 π] / 30], Specularity[White, 100]},
    MaxRecursion → qual], {k, 0, 2 π - π / 6, π / 6}];
bottoms = Table[ParametricPlot3D[TorusPoints3[θ, α^3 - k, F[α^3]],
    {θ, π / 2, 3 π / 2}, {α, 0, π^(1/3)}, Mesh → None,
    PlotStyle → {Hue[Mod[k - 5 π / 3, 2 π] / 30 + 1 / 4], Specularity[White, 100]},
    MaxRecursion → qual], {k, 0, 2 π - π / 6, π / 6}];
Ctorus3 = ParametricPlot3D[TorusPoints3[θ, α, 1], {θ, 0, 2 π}, {α, 0, 2 π},
   Mesh → None, PlotStyle → {Blue, Specularity[White, 100]}, MaxRecursion → qual,
   RegionFunction → Function[{x, y, z}, Max[Abs /@ {x, y, z}] < 20 && z < 1.5]];
yellow = ParametricPlot3D[TorusPoints3[θ^3 - 4 π / 3, α, 1 / F[θ^3]],
   {θ, 0, (2 π)^(1/3)}, {α, 0, 2 π}, Mesh → None,
   PlotStyle → {Hue[Mod[4 π / 3 - 3 π / 2, 2 π] / 30], Specularity[White, 100]},
   MaxRecursion → qual, RegionFunction → Function[{x, y, z}, z < 1.5]];
cyan = ParametricPlot3D[TorusPoints3[θ, α^3 - 3 π / 2, F[α^3]], {θ, 0, 2 π}, {α, 0, (2 π)^(1/3)},
   Mesh → None, PlotStyle → {Hue[(11 π / 6) / 30 + 1 / 4], Specularity[White, 100]},
   MaxRecursion → qual, RegionFunction → Function[{x, y, z}, z < 1.5]];
back = ParametricPlot3D[TorusPoints3[θ, α^3 - 11 π / 6, F[α^3]],
   {θ, 0, 2 π}, {α, (0.39 π)^(1/3), (3 π / 2)^(1/3)}, Mesh → None,
   PlotStyle → {Hue[(π / 6) / 30 + 1 / 4], Specularity[White, 100]},
   MaxRecursion → qual, RegionFunction → Function[{x, y, z}, z < 50]];
```

```
Show[{tops, bottoms, Ctorus3, yellow, cyan, back},
  PlotRange → {{-5, 20}, {-20, 5}, {-20, 10}}, Boxed → False, Axes → False,
  Background → Gray, ImageSize → {1296, 768}, ViewVector → {{12, 8, 32}, {0.8, 0, 0}},
  ViewAngle → π/24, ViewVertical → {0, 1, 0}] // Rasterize
```